



Techniques of Water-Resources Investigations  
of the United States Geological Survey

Chapter A1

**A MODULAR THREE-DIMENSIONAL  
FINITE-DIFFERENCE GROUND-WATER  
FLOW MODEL**

By Michael G. McDonald and  
Arlen W. Harbaugh

This chapter supersedes U.S. Geological  
Survey Open-File Report 83-875

Book 6

MODELING TECHNIQUES

### Narrative for Module DRN1RP

This module reads data to build the drain list.

1. Read ITMP. ITMP is the number of drains or a flag indicating that drain data from the previous stress period should be reused.

2. Test ITMP. If ITMP is less than zero, the drain data read for the last stress period will be reused. Print a message to that effect and RETURN.

3. If ITMP is greater than or equal to zero, it is the number of drains for this stress period. Set the number of drains (NDRAIN) in the current stress period equal to ITMP.

4. Compare the number of drains (NDRAIN) in the current stress period to the number specified as the maximum for the simulation (MXDRN). If NDRAIN is greater than MXDRN, STOP.

5. Print the number of drains in the current stress period (NDRAIN).

6. See if there are any drains. If there are no drains in the current stress period (NDRAIN = 0), bypass further drain processing.


7. Read and print the layer, row, column, elevation, and conductance for each drain.

8. RETURN.

Flow Chart for Module DRN1RP

ITMP is both a flag and a counter. If it is greater than or equal to zero, it is the number of drains to be simulated during the current stress period. If it is less than zero, it indicates that the drains simulated in the last stress period should be simulated in the current stress period.

MXDRN is the maximum number of drains to be simulated.



```

SUBROUTINE DRN1RP(DRAI,NDRAIN,MXDRN,IN,IOUT)
C
C
C-----VERSION 1603 25APR1983 DRN1RP
C *****
C READ DRAIN LOCATIONS, ELEVATIONS, AND CONDUCTANCES
C *****
C
C SPECIFICATIONS:
C -----
C DIMENSION DRAI(5,MXDRN)
C -----
C1-----READ ITMP(NUMBER OF DRAIN CELLS OR FLAG TO REUSE DATA)
      READ(IN,8) ITMP
      8 FORMAT(I10)
C
C2-----TEST ITMP
      IF(ITMP.GE.0) GO TO 50
C
C2A-----IF ITMP<0 THEN REUSE DATA FROM LAST STRESS PERIOD.
      WRITE(IOUT,7)
      7 FORMAT(1H0,'REUSING DRAINS FROM LAST STRESS PERIOD')
      RETURN
C
C3-----IF ITMP=>0 THEN IT IS THE NUMBER OF DRAINS.
      50 NDRAIN=ITMP
      IF(NDRAIN.LE.MXDRN) GO TO 100
C
C4-----IF NDRAIN>MXDRN THEN STOP
      WRITE(IOUT,99) NDRAIN,MXDRN
      99 FORMAT(1H0,'NDRAIN(',I4,') IS GREATER THAN MXDRN(',I4,')')
      STOP
C
C5-----PRINT NUMBER OF DRAINS IN THIS STRESS PERIOD.
      100 WRITE(IOUT,1) NDRAIN
      1 FORMAT(1H0, '//1X, I5, ' DRAINS')
C
C6-----IF THERE ARE NO DRAINS THEN RETURN.
      IF(NDRAIN.EQ.0) GO TO 260
C
C7-----READ AND PRINT DATA FOR EACH DRAIN.
      WRITE(IOUT,3)
      3 FORMAT(1H0,15X,'LAYER',5X,'ROW',5X
      1,'COL ELEVATION CONDUCTANCE DRAIN NO. '/1X,15X,60(' -'))
      DO 250 II=1,NDRAIN
      READ (IN,4) K,I,J,DRAI(4,II),DRAI(5,II)
      4 FORMAT(3I10,2F10.0)
      WRITE (IOUT,5) K,I,J,DRAI(4,II),DRAI(5,II),II
      5 FORMAT(1X,15X,I4,I9,I8,G13.4,G14.4,I8)
      DRAI(1,II)=K
      DRAI(2,II)=I
      DRAI(3,II)=J
      250 CONTINUE
C
C8-----RETURN
      260 RETURN
C
      END

```

List of Variables for Module DRN1RP

<u>Variable</u>	<u>Range</u>	<u>Definition</u>
DRAI	Package	DIMENSION (5,MXDRN), For each drain: layer, row, column, head in drain, and conductance into drain.
I	Module	Index for rows.
II	Module	Index for drains.
IN	Package	Primary unit number from which input for this package will be read.
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
ITMP	Module	Flag or number of drains.  $\geq 0$ , number of drains active during the current stress period. $< 0$ , same drains active during the last stress period will be active during the current stress period.
J	Module	Index for columns.
K	Module	Index for layers.
MXDRN	Package	Maximum number of drains active at any one time.
NDRAIN	Package	Number of drains active during the current stress period.

### Narrative for Module DRN1FM


This module adds terms representing drain leakage to the accumulators HCOF and RHS.

1. If NDRAIN is less than or equal to zero in the current stress period, there are no drains. RETURN.
2. For each drain in the drain list, DO STEPS 3-7.
3. Determine the column (IC), row (IR), and layer (IL).
4. If the cell is external ( $IBOUND(IC, IR, IL) \leq 0$ ), bypass processing on this drain and go on to the next drain.
5. If the cell is internal, get the drain data (elevation and conductance).
6. If the head in the aquifer (HHNEW) is greater than the elevation of the drain, there is no drain leakage. RETURN.
7. If the head in the aquifer (HHNEW) is greater than the elevation of the drain (EL), add the term  $-C*EL$  (C is the drain conductance) to the accumulator RHS and the term  $-C$  to the accumulator HCOF.
8. RETURN.

Flow Chart for Module DRN1FM

RHS is an accumulator in which the right hand side of the equation is formulated.

HCOF is an accumulator in which the coefficient of head in the cell is formulated.



```

SUBROUTINE DRN1FM(NDRAIN,MXDRN,DRAI,HNEW,HCOF,RHS,IBOUND,
1          NCOL,NROW,NLAY)
C
C-----VERSION 1030 10APR1985 DRN1FM
C
C *****
C ADD DRAIN FLOW TO SOURCE TERM
C *****
C
C SPECIFICATIONS:
C -----
C DOUBLE PRECISION HNEW
C
C DIMENSION DRAI(5,MXDRN),HNEW(NCOL,NROW,NLAY),
1          RHS(NCOL,NROW,NLAY),IBOUND(NCOL,NROW,NLAY),
1          HCOF(NCOL,NROW,NLAY)
C -----
C
C1-----IF NDRAIN<=0 THERE ARE NO DRAINS. RETURN
          IF(NDRAIN.LE.0) RETURN
C
C2-----PROCESS EACH CELL IN THE DRAIN LIST
          DO 100 L=1,NDRAIN
C
C3-----GET COLUMN, ROW AND LAYER OF CELL CONTAINING DRAIN.
          IL=DRAI(1,L)
          IR=DRAI(2,L)
          IC=DRAI(3,L)
C
C4-----IF THE CELL IS EXTERNAL SKIP IT.
          IF(IBOUND(IC,IR,IL).LE.0) GO TO 100
C
C5-----IF THE CELL IS INTERNAL GET THE DRAIN DATA.
          EL=DRAI(4,L)
          HHNEW=HNEW(IC,IR,IL)
C
C6-----IF HEAD IS LOWER THAN DRAIN THEN SKIP THIS CELL.
          IF(HHNEW.LE.EL) GO TO 100
C
C7-----HEAD IS HIGHER THAN DRAIN. ADD TERMS TO RHS AND HCOF.
          C=DRAI(5,L)
          HCOF(IC,IR,IL)=HCOF(IC,IR,IL)-C
          RHS(IC,IR,IL)=RHS(IC,IR,IL)-C*EL
          100 CONTINUE
C
C8-----RETURN
          RETURN
          END

```



List of Variables for Module DRN1FM

<u>Variable</u>	<u>Range</u>	<u>Definition</u>
C	Module	Conductance into the drain.
DRAI	Package	DIMENSION (5,MXDRN), For each drain: layer, row, column, head in the drain and conductance into the drain.
EL	Module	Elevation of the drain (head in the drain).
HCOF	Global	DIMENSION (NCOL,NROW,NLAY), Coefficient of head in the cell (J,I,K) in the finite-difference equation.
HHNEW	Module	Head in the cell containing the drain.
HNEW	Global	DIMENSION (NCOL,NROW,NLAY), Most recent estimate of head in each cell. HNEW changes at each iteration.
IBOUND	Global	DIMENSION (NCOL,NROW,NLAY), Status of each cell. < 0, constant-head cell = 0, inactive cell > 0, variable-head cell
IC	Module	Index for columns.
IL	Module	Index for layers.
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
IR	Module	Index for rows.
MXDRN	Package	Maximum number of drains active at any one time.
NCOL	Global	Number of columns in the grid.
NDRAIN	Package	Number of drains active during the current stress period.
NLAY	Global	Number of layers in the grid.
NROW	Global	Number of rows in the grid.
RHS	Global	DIMENSION (NCOL,NROW,NLAY), Right hand side of the finite-difference equation. RHS is an accumulation of terms from several different packages.

## Narrative for Module DRN1BD

This module calculates rates and volumes transferred between the aquifer and drains.

1. Initialize the cell-by-cell flow-term flag (IBD) and the rate accumulator (RATOUT).

2. If there are no drains ( $NDRAIN \leq 0$ ), skip down to step 12 and put zeros into the budget terms for drains.

3. Test to see if cell-by-cell flow terms are to be saved on disk. They will not be saved if either of the following conditions hold: (1) this is not the proper time step ( $ICBCFL = 0$ ) or (2) cell-by-cell flow terms are not needed for drains during this simulation ( $IDRNCB \leq 0$ ). If cell-by-cell flow terms will be saved for drains, set the cell-by-cell flow-term flag (IBD) and clear the buffer in which they will be accumulated (BUFF).

4. For each drain, do steps 3-11 accumulating flows into drains.

5. Determine the row, column, and layer of the cell containing the drain.

6. If the cell is external ( $IBOUND(I,J,K) \leq 0$ ), bypass further processing of this drain.

7. Get the drain parameters from the drain list.

8. If the head in the cell is less than the elevation of the drain, bypass further processing of this drain.

9. If the head in the cell is greater than the elevation of the drain, set "Q" equal to the conductance of the drain (C) times the drain elevation (EL) minus the head in the cell (HHNEW) ( $Q = C*(EL - HHNEW)$ ). Add Q to the accumulator RATOUT to get the total flow from the aquifer into drains.

10. If the cell-by-cell flow terms are to be printed ( $IDRNCB < 0$  and  $ICBCFL \neq 0$ ), print Q.

11. If the cell-by-cell flow terms for drains are to be saved, add Q to the buffer (BUFF).

12. See if the cell-by-cell flow terms are to be saved ( $IBD = 1$ ). If they are, call module UBUDSV to record the buffer (BUFF) onto disk.

13. Move RATOUT into the VBVL array for printing by BAS10T. Add RATOUT multiplied by the time-step length to the volume accumulator in VBVL for printing by BAS10T. Move the drain budget-term labels to VBNM for print by BAS10T.

14. Increment the budget-term counter (MSUM). See the section in the Basic Package for a detailed explanation of VBVL, VBNM, and MSUM.

15. RETURN.

## Flow Chart for Module DRN1BD

IBD is a flag which, if set, causes cell-by-cell flow terms for drains to be recorded.

EXTERNAL: a cell is said to be external if it is either no flow or constant head (i.e., an equation is not formulated for the cell).

BUFFER is an array in which values are stored as they are being gathered for printing or recording.

RATOUT is an accumulator to which all flows out of the aquifer are added.

Q is the negative of discharge to a drain.

EL is the elevation of the drain.

IDRNCB is a flag and a unit number.


If  $IDRNCB > 0$ , it is the unit number on which cell-by-cell flow terms will be recorded whenever ICBCFL is set.

If  $IDRNCB = 0$ , cell-by-cell flow terms will not be printed or recorded.

If  $IDRNCB < 0$ , drain leakage for each drain will be printed whenever ICBCFL is set.

ICBCFL is a flag.

If  $ICBCFL \neq 0$ , cell-by-cell flow terms will be either printed or recorded for the current time step.



```

SUBROUTINE DRN1BD(NDRAIN,MXDRN,VBNM,VBVL,MSUM,DRAI,DELT,HNEW,
1          NCOL,NROW,NLAY,IBOUND,KSTP,KPER,IDRNCB,ICBCFL,BUFF,IOUT)
C
C-----VERSION 1338 22AUG1987 DRN1BD
C
C          *****
C          CALCULATE VOLUMETRIC BUDGET FOR DRAINS
C          *****
C
C          SPECIFICATIONS:
C          -----
C          CHARACTER*4 VBNM,TEXT
C          DOUBLE PRECISION HNEW
C
C          DIMENSION VBNM(4,MSUM),VBVL(4,MSUM),DRAI(5,MXDRN),
1          HNEW(NCOL,NROW,NLAY),IBOUND(NCOL,NROW,NLAY),
2          BUFF(NCOL,NROW,NLAY)
C          DIMENSION TEXT(4)
C
C          DATA TEXT(1),TEXT(2),TEXT(3),TEXT(4) /' ',' ','DR','AINS'/
C          -----
C1-----INITIALIZE CELL-BY-CELL FLOW TERM FLAG (IBD) AND
C1-----ACCUMULATORS (RATIN AND RATOUT).
          RATOUT=0.
          IBD=0
C
C2-----IF THERE ARE NO DRAINS THEN DO NOT ACCUMULATE DRAIN FLOW
          IF(NDRAIN.LE.0) GO TO 200
C
C3-----TEST TO SEE IF CELL-BY-CELL FLOW TERMS ARE NEEDED.
          IF(ICBCFL.EQ.0 .OR. IDRNCB.LE.0) GO TO 60
C
C3B-----CELL-BY-CELL FLOW TERMS ARE NEEDED SET IBD AND CLEAR BUFFER.
          IBD=1
          DO 50 IL=1,NLAY
          DO 50 IR=1,NROW
          DO 50 IC=1,NCOL
          BUFF(IC,IR,IL)=0.
          50 CONTINUE
C
C4-----FOR EACH DRAIN ACCUMULATE DRAIN FLOW
          60 DO 100 L=1,NDRAIN
C
C5-----GET LAYER, ROW & COLUMN OF CELL CONTAINING REACH.
          IL=DRAI(1,L)
          IR=DRAI(2,L)

```

```

        IC=DRAI(3,L)
C
C6-----IF CELL IS EXTERNAL IGNORE IT.
        IF(IBOUND(IC,IR,IL).LE.0) GO TO 100
C
C7-----GET DRAIN PARAMETERS FROM DRAIN LIST.
        EL=DRAI(4,L)
        C=DRAI(5,L)
        HHNEW=HNEW(IC,IR,IL)
C
C8-----IF HEAD LOWER THAN DRAIN THEN FORGET THIS CELL.
        IF(HHNEW.LE.EL) GO TO 100
C
C9-----HEAD HIGHER THAN DRAIN.  CALCULATE Q=C*(EL-HHNEW).
C9-----SUBTRACT Q FROM RATOUT.
        Q=C*(EL-HHNEW)
        RATOUT=RATOUT-Q
C
C10-----PRINT THE INDIVIDUAL RATES IF REQUESTED(IDRNCB<0).
        IF(IDRNCB.LT.0.AND.ICBCFL.NE.0) WRITE(IOUT,900) (TEXT(N),N=1,4),
1      KPER,KSTP,L,IL,IR,IC,Q
        900 FORMAT(1H0,4A4,' PERIOD',I3,' STEP',I3,' DRAIN',I4,
1      ' LAYER',I3,' ROW',I4,' COL',I4,' RATE',G15.7)
C
C11-----IF C-B-C FLOW TERMS ARE TO BE SAVED THEN ADD Q TO BUFFER.
        IF(IBD.EQ.1) BUFF(IC,IR,IL)=BUFF(IC,IR,IL)+Q
        100 CONTINUE
C
C12-----IF C-B-C FLOW TERMS WILL BE SAVED CALL UBUDSV TO RECORD THEM.
        IF(IBD.EQ.1) CALL UBUDSV(KSTP,KPER,TEXT,IDRNCB,BUFF,NCOL,NROW,
1      NLAY,IOUT)
C
C13-----MOVE RATES,VOLUMES & LABELS INTO ARRAYS FOR PRINTING.
        200 VBVL(3,MSUM)=0.
        VBVL(4,MSUM)=RATOUT
        VBVL(2,MSUM)=VBVL(2,MSUM)+RATOUT*DELT
        VBNM(1,MSUM)=TEXT(1)
        VBNM(2,MSUM)=TEXT(2)
        VBNM(3,MSUM)=TEXT(3)
        VBNM(4,MSUM)=TEXT(4)
C
C14-----INCREMENT BUDGET TERM COUNTER
        MSUM=MSUM+1
C
C15-----RETURN
        RETURN
        END

```

List of Variables for Module DRN1BD

<u>Variable</u>	<u>Range</u>	<u>Definition</u>
BUFF	Global	DIMENSION (NCOL,NROW,NLAY), Buffer used to accumulate information before printing or recording it.
C	Module	Conductance into drains.
DELT	Global	Length of the current time step.
DRAI	Package	DIMENSION (5,MXDRN), For each drain: layer, row, column, head in the drain and conductance into the drain.
EL	Module	Elevation of the drain (head in the drain).
HHNEW	Module	Head in the cell containing the drain.
HNEW	Global	DIMENSION (NCOL,NROW,NLAY), Most recent estimate of head in each cell. HNEW changes at each iteration.
IBD	Package	Flag. = 0, cell-by-cell flow terms for this package will not be recorded. ≠ 0, cell-by-cell flow terms for this package will be recorded.
IBOUND	Global	DIMENSION (NCOL,NROW,NLAY), Status of each cell. < 0, constant-head cell = 0, inactive cell > 0, variable-head cell
IC	Module	Index for columns.
ICBCFL	Global	Flag. = 0, cell-by-cell flow terms will not be recorded or printed for the current time step. ≠ 0, cell-by-cell flow terms will be recorded for the current time step.
IDRNCB	Package	Flag. > 0 and if ICBCFL ≠ 0, cell-by-cell flow terms for the DRN1 Package will be recorded on UNIT = IDRNCB.
IL	Module	Index for layers.
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
IR	Module	Index for rows.
KPER	Global	Stress period counter.

List of Variables for Module DRN1BD (Continued)

<u>Variable</u>	<u>Range</u>	<u>Definition</u>
KSTP	Global	Time step counter. Reset at the start of each stress period.
L	Module	Index for drains.
MSUM	Global	Counter for budget entries and labels in VBVL and VBNM.
MXDRN	Package	Maximum number of drains active at any one time.
NCOL	Global	Number of columns in the grid.
NDRAIN	Package	Number of drains active during the current stress period.
NLAY	Global	Number of layers in the grid.
NROW	Global	Number of rows in the grid.
Q	Module	Flow from a drain into a cell. (Reverse the sign to get the flow into the drain.)
RATOUT	Module	Accumulator for the total flow out of the flow field into the drains.
TEXT	Module	Label to be printed or recorded with the array data.
VBNM	Global	DIMENSION (4,20), Labels for entries in the volumetric budget.
VBVL	Global	DIMENSION (4,20), Entries for the volumetric budget. For flow component N, the values in VBVL are: (1,N) Rate for the current time step into the flow field. (2,N) Rate for the current time step out of the flow field. (3,N) Volume into the flow field during simulation. (4,N) Volume out of the flow field during simulation.



CHAPTER 10

EVAPOTRANSPIRATION PACKAGE

Conceptualization and Implementation

The Evapotranspiration (ET) Package simulates the effects of plant transpiration and direct evaporation in removing water from the saturated ground water regime. The approach is based on the following assumptions: (1) when the water table is at or above a specified elevation, termed the "ET surface" in this report, evapotranspiration loss from the water table occurs at a maximum rate specified by the user; (2) when the depth of the water table below the ET surface elevation exceeds a specified interval, termed the "extinction depth" or "cutoff depth" in this report, evapotranspiration from the water table ceases; and (3) between these limits, evapotranspiration from the water table varies linearly with water table elevation.

This can be expressed in equation form as

$$R_{ETi,j} = R_{ETMi,j} \quad h_{i,j,k} > h_{Si,j} \quad (71)$$

$$R_{ETi,j} = 0 \quad h_{i,j,k} < h_{Si,j} - d_{i,j} \quad (72)$$

$$R_{ETi,j} = R_{ETMi,j} \left\{ \frac{h_{i,j,k} - (h_{Si,j} - d_{i,j})}{d_{i,j}} \right\} \quad (h_{Si,j} - d_{i,j}) \leq h_{i,j,k} \leq h_{Si,j} \quad (73)$$

where  $R_{ETi,j}$  is the rate of loss per unit surface area of water table due to evapotranspiration, in volume of water per unit area per unit time, within the map area  $DEL R_j DEL C_i$ ;  $h_{i,j,k}$  is the head, or water table elevation in the cell from which the evapotranspiration occurs;  $R_{ETMi,j}$  is the maximum possible value of  $R_{ETi,j}$ ;  $h_{Si,j}$  is the ET surface elevation, or the water table elevation at which this maximum value of evapotranspiration loss occurs; and  $d_{i,j}$  is the cutoff or extinction depth, such that when the

distance between  $h_{si,j}$  and  $h_{i,j,k}$  exceeds  $d_{i,j}$  evapotranspiration ceases.

In implementing the finite difference approach the volumetric rate of evapotranspiration loss from a given cell is required. This is given as the product of the loss rate per unit area, and the horizontal surface area,  $DEL R_j DEL C_i$ , of the cell from which the loss occurs, i.e.

$$Q_{ETi,j} = R_{ETi,j} * DEL R_j * DEL C_i \quad (74)$$

where  $Q_{ETi,j}$  is the evapotranspiration, in volume of water per unit time, through the area  $DEL R_j DEL C_i$ . If the maximum value of  $Q_{ETi,j}$  (corresponding to  $R_{ETMi,j}$ ) is designated  $Q_{ETMi,j}$ , equations (71)-(73) can be expressed in terms of volumetric discharge as

$$Q_{ETi,j} = Q_{ETMi,j} \quad h_{i,j,k} > h_{si,j} \quad (75)$$

$$Q_{ETi,j} = 0 \quad h_{i,j,k} < h_{si,j} - d_{i,j} \quad (76)$$

$$Q_{ETi,j} = Q_{ETMi,j} \left\{ \frac{h_{i,j,k} - (h_{si,j} - d_{i,j})}{d_{i,j}} \right\} \quad (h_{si,j} - d_{i,j}) \leq h_{i,j,k} \leq h_{si,j} \quad (77)$$

Figure 42 shows a graph of evapotranspiration loss,  $Q_{ETi,j}$ , vs head in cell  $i,j,k$  based on equations (75)-(77). Comparison of the ET function with the river or drain functions shows that the three are mathematically similar, except that the linear portion of the ET function is bounded at both ends by constant values, rather than only at the lower end.

Evapotranspiration is drawn from only one cell in the vertical column beneath the map area  $DEL R_j * DEL C_i$ ; the user designates the cell (i.e. the layer, k) using one of two options. Under the first option, evapotranspiration is always drawn from the uppermost layer of the model; under the second, the user specifies the cell, within the vertical column at  $i,j$ , from which the evapotranspiration is to be taken. In either case the computed evapotrans-




Figure 42.—Plot of volumetric evapotranspiration,  $Q_{ET}$ , as a function of head,  $h$ , in a cell where  $d$  is the cutoff depth and  $h_s$  is the ET surface elevation.

piration has no influence on the simulation if the designated cell is either a no-flow cell or a constant head cell.

For each cell location,  $(i, j)$ , in the horizontal plane, and for each stress period (unless an option is exercised to use prior values) the ET package reads values of  $R_{ETM}$  (maximum evapotranspiration loss per unit area per unit time) into an array labelled EVTR. These rates are immediately multiplied by cell areas,  $DEL R_j * DEL C_i$ , to obtain the maximum volumetric rate of evapotranspiration from each cell,  $Q_{ETM}$ ; these maximum volumetric rates then replace the values of  $R_{ETM_{i,j}}$  in the EVTR array. Thus, the input to the EVTR array consists of maximum evapotranspiration rates per unit area, and as such must have dimensions  $Lt^{-1}$ . In the calculation carried out within the model, however, the entries in the EVTR array appear as maximum volumetric rates, having dimensions  $L^3t^{-1}$ .

Values of  $h_{si,j}$ , the ET surface elevation (or water table elevation at which evapotranspiration is maximum), are read into the two dimensional array SURF by the ET package; values of the cutoff depth or extinction depth are read into the two-dimensional array EXDP. Because the term  $Q_{ET_{i,j}}$  of equations (75)-(77) has been defined as an outflow from the aquifer it must be subtracted from the left side of equation (24). In terms of the expressions HCOF and RHS of equation (26), this is accomplished in the ET package as follows:

- 1) if  $h_{i,j,k} < (h_{si,j,k} - d_{i,j})$  no changes are made in the terms HCOF or RHS for cell  $i,j,k$ ;
- 2) if  $h_{i,j,k} > h_{si,j}$ ,  $Q_{ETM_{i,j}}$  is added to  $RHS_{i,j}$ ; and
- 3) if  $(h_{si,j,k} - d_{i,j}) \leq h_{i,j,k} \leq h_{si,j,k}$ ,  $-Q_{ETM_{i,j}}/d_{i,j}$  is added to


HCOF $_{i,j,k}$  and  $-Q_{ETM_{i,j}} \left\{ \frac{h_{si,j} - d_{i,j}}{d_{i,j}} \right\}$  is added to RHS  $i,j,k$ .

The value of  $h_{s,i,j}$ , the water table elevation at which evapotranspiration is maximum, should normally be taken as the average land surface elevation in the map area  $DEL R_j DEL C_i$ ; the cutoff or extinction depth,  $d_{i,j}$ , is then frequently assumed to be on the order of six to eight feet below land surface (although considerable variation can be introduced by climatic factors, the presence of deep-rooted phreatophytes, or so on). Where the distance from land surface to the water table varies extensively within the area of a cell, care must be exercised in implementing the ET package and in choosing the various parameters of equation (70), or misleading results may be obtained.


The options for selection of the layer from which ET is to be drawn provide some flexibility in adapting the package to special situations, but also require some care in implementation. Figure 43 shows a situation similar to that discussed for the recharge package, in which a cross sectional model has been progressively truncated to follow the water table, using the provision for horizontal conductance formulation under water table conditions (Chapter 5). Figure 43-a shows the hydrologic situation under study, and figure 43-b the final distribution of variable head and inactive (no flow) cells obtained in the simulation.

Under option 1 (figure 43-c), evapotranspiration is drawn only from the uppermost layer of the model; in the problem shown, the presence of no flow cells in this layer deletes evapotranspiration from the right half of the model, so that the simulation fails to represent field conditions.

Figure 43-d shows the situation which could be achieved through the use of option 2, assuming that the simulation was carried out in stages and




a



Status of Cells at End of Simulated Period

b


- Variable Head Cell
- Constant Head Cell
- Inactive Cell



Cells from Which ET Is Abstracted Under Option 1

c

- Cell from Which ET Is Abstracted



Cells from Which ET Is Abstracted Under Option 2

d

Layer Indicators Specified in the IEVT Array

- Cell from Which ET Is Abstracted

Figure 43.—Hypothetical problem showing cells from which ET will be abstracted under the two options available in the ET Package.

that the user interacted with the simulation process, designating the cells from which evapotranspiration was to be drawn as the truncation of the mesh developed.

## Evapotranspiration Package Input

Input to the Evapotranspiration (EVT) Package is read from the unit specified in IUNIT (5).

FOR EACH SIMULATION

### EVTIAL

1. Data: NEVTOP      IEVTCB  
Format: I10        I10

FOR EACH STRESS PERIOD

### EVTIRP

2. Data:    INSURF      INEVTR      INEXDP      INIEVT  
Format:    I10        I10        I10        I10
3. Data:    SURF  
Module:    U2DREL
4. Data:    EVTR  
Module:    U2DREL
5. Data:    EXDP  
Module:    U2DREL

IF THE ET OPTION IS EQUAL TO TWO

6. Data:    IEVT  
Module:    U2DINT

### Explanation of Fields Used in Input Instructions

NEVTOP--is the evapotranspiration (ET) option code. ET parameters (ET surface, maximum ET rate, and extinction depth) are specified in two-dimensional arrays, SURF, EVTR, and EXDP, with one value for each vertical column. Accordingly, ET is calculated for one cell in each vertical column. The option codes determine for which cell in the column ET will be calculated.

- 1 - ET is calculated only for cells in the top grid layer.
- 2 - The cell for each vertical column is specified by the user in array IEVT.



IEVTCB--is a flag and a unit number.

If  $IEVTCB > 0$ , it is the unit number on which cell-by-cell flow terms will be recorded whenever ICBCFL (see Output Control) is set.

If  $IEVTCB \leq 0$ , cell-by-cell flow terms will not be printed or recorded.

INSURF--is the ET surface (SURF) read flag.

If  $INSURF \geq 0$ , an array containing the ET surface elevation will be read.

If  $INSURF < 0$ , the ET surface from the preceding stress period will be reused.

INEVTR--is the maximum ET rate (EVTR) read flag.

If  $INEVTR \geq 0$ , an array containing the maximum ET rate will be read.

If  $INEVTR < 0$ , the maximum ET rate from the preceding stress period will be reused.

INEXDP--is the extinction depth (EXDP) read flag.

If  $INEXDP \geq 0$ , an array containing the extinction depth (EXDP) will be read.

If  $INEXDP < 0$ , the extinction depth from the preceding stress period will be reused.

INIEVT--is the layer indicator (IEVT) read flag. It is used only if the ET option (NEVTOP) is equal to two.

If  $INIEVT \geq 0$ , an array containing the layer indicators (IEVT) will be read.

If  $INIEVT < 0$ , layer indicators used during the preceding stress period will be reused.

SURF--is the elevation of the ET surface.

EVTR--is the maximum ET rate (volume of water per unit area ( $Lt^{-1}$ )).

EXDP--is the ET extinction depth.

IEVT--is the layer indicator array. For each horizontal location, it indicates the layer from which ET is removed. It is needed only if the ET option is equal to two.

SAMPLE INPUT TO THE EVAPOTRANSPIRATION PACKAGE USING ET OPTION 1

DATA ITEM	EXPLANATION	INPUT RECORDS
1	{NEVTOP, IEVTCH}	1 0
2	Stress period 1--{INSURF, INEVTR, INEXDP, INIEVT}	1 1 (10F5.0)
3	Control record for ET surface array	27 710 715 720 725 730 735
	ET surface	715 720 725 730 735 740
		720 725 730 735 740 745
		725 730 735 740 745 750
		730 735 740 745 750 755
4	Control record for maximum ET rate	0 9.65E-7
5	Control record for extinction depth array	0 10.
2	Stress period 2--{INSURF, INEVTR, INEXDP, INIEVT}	-1 -1
4	Control record for maximum ET rate	0 8.23E-7
2	Stress period 3--{INSURF, INEVTR, INEXDP, INIEVT}	-1 -1
4	Control record for maximum ET rate	27 9.65E-7 (10F4.0)
	max ET rate	1.2 1.2 1.2 1.2 1.3
		1.2 1.2 1.2 1.3 1.4
		1.2 1.2 1.3 1.4 1.4
		1.0 1.0 1.0 1.1 1.1
		1.2 1.3 1.3 1.4 1.4
		1.3 1.3 1.4 1.4 1.4

SAMPLE INPUT TO THE EVAPOTRANSPIRATION PACKAGE USING ET OPTION 2

DATA ITEM	EXPLANATION	INPUT RECORDS
1	{NEVTOP, IEVTCH}	2 45
2	Stress period 1--{INSURF, INEVTR, INEXDP, INIEVT}	1 1 (10F5.0)
3	Control record for ET surface array	27 710 715 720 725 730 735
	ET surface	715 720 725 730 735 740
		720 725 730 735 740 745
		725 730 735 740 745 750
		730 735 740 745 750 755
4	Control record for maximum ET rate	0 9.65E-7
5	Control record for extinction depth array	0 10.
6	Control record for layer indicator array	12 1 (20I2)
	Layer numbers	1 2 2 2 3
		1 2 2 2 2
		1 1 2 2 2
		1 1 1 1 2
		1 1 1 1 1
		1 1 1 1 1
2	Stress period 2--{INSURF, INEVTR, INEXDP, INIEVT}	1 1 1 1 1
4	Control record for maximum ET rate	-1 -1 8.23E-7

FIELDS IN ARRAY CONTROL RECORDS ARE---{ LOCAT, CONST, FMTIN, IPRN}